

·继续教育园地·

重复测量数据的常用统计分析方法

冯国双

国家儿童医学中心 首都医科大学附属北京儿童医院大数据和工程研究中心, 北京 100045

通信作者:冯国双, Email: glxfqsh@163.com

【摘要】 重复测量数据是医学中常见的一种数据,该数据的分析不能简单对每个时间点分别比较,而应采用专业的统计分析方法。本文介绍了目前重复测量数据常见的三种方法,即重复测量方差分析、广义估计方程和多水平模型,并基于实例对三种方法的具体软件实现、结果解读进行了详细解释,同时对三种方法实际应用的情形进行了分析比较,以期临床科研人员能够正确分析重复测量数据,提高数据的分析效率。

【关键词】 模型; 统计学; 数据说明; 统计; 重复测量数据

Statistical methods for repeated measurement data in scientific research

Feng Guoshuang

Big Data and Engineering Research Center, Beijing Children's Hospital, Capital Medical University/
National Center for Children's Health, Beijing 100045, China

Corresponding author: Feng Guoshuang, Email: glxfqsh@163.com

【Abstract】 Repeated measurement data is a common type of data in medicine, which can not be simply compared at each time point, and a professional statistical analysis method should be used to analysis this kind of data. Three common statistical methods were introduced for repeated measurement data, including repeated measurement analysis of variance, generalized estimation equations and multilevel models. The implementation of specific software and related results for the three methods based on some cases were also explained in the article. Additionally, we compared the actual application of the three methods, in order to help clinical researchers to analyze repeated measurement data correctly and to improve their efficiency of data analysis.

【Key words】 Models, statistical; Data interpretation; Statistical repeated measures

重复测量数据是指对同一个体在不同时间点的测量,这种数据在医学研究中较为常见,比较典型的数据形式如对一组人群分别在干预前后不同的时间点观察其结局情况;或将一组人群分配至不同组别,对每组人群分别在干预前后不同的时间点观察其结局。重复测量数据的个体观测值不完全独立,数据间存在趋同性,如果采用独立数据的统计推断方法进行分析,往往会增大第 I 类错误发生的概率。基于不同的研究目的和分析策略,重复测量数据可采用不同的分析方法,医学中常见的有重复测量方差分析、广义估计方程、多水平模型。本研究主要从重复测量数据的分析

思路入手,介绍各分析方法的思想和过程,并对几种方法进行比较。

一、重复测量方差分析

(一)重复测量方差分析思路

重复测量方差分析仍是属于方差分析的范畴,因此其思路也是基于方差分解。其与单变量方差分析思路的不同之处在于:单变量方差分析是对某一变量的方差进行分解,而重复测量数据存在多个时间点的测量结果,并不仅有 1 个变量,而有多个变量,从而形成多个变量的方差-协方差矩阵。因此,重复测量方差分析不是对 1 个变量的方差进行分

DOI: 10.3760/cma.j.cn112150-20200514-00728

收稿日期 2020-05-14 本文编辑 张振伟

引用本文:冯国双. 重复测量数据的常用统计分析方法[J]. 中华预防医学杂志, 2020, 54(7): 804-812. DOI: 10.3760/cma.j.cn112150-20200514-00728.



解,而是对多个变量的方差-协方差矩阵进行分解。

但从方差或方差-协方差矩阵的分解结构来看,二者仍是一致的。单变量方差分析是将总的离均差平方和(SS_{total})分解为模型解释的变异(SS_{model} ,通常指组别因素)和误差变异(SS_{error})。重复测量方差分析则是将总的交叉乘积平方和($SSCP$)矩阵 $T(SSCP_T)$ 分解为模型 $SSCP$ 矩阵 $H(SSCP_H)$ 和误差 $SSCP$ 矩阵 $E(SSCP_E)$ ^[1]。

(二)重复测量方差分析的用途

从实际应用来看,重复测量方差分析可用于以下目的:比较组间有无差异;比较各时间点间有无差异;比较组间的时间变化趋势有无差异(即随时间变化的多条曲线是否平行)。

(三)重复测量方差分析的软件实现

例1:某药物开展一项治疗急性脑梗死的研究,将20名急性脑梗死患者随机分为两组,每组10例,分别接受试验药和对照药,于治疗前(0周)、治疗后1、2、3、4周,分别以某症状评分指标Y作为治疗效果指标。欲实现以下目的:(1)两组的Y值差异是否具有统计学意义;(2)不同时间点的Y值差异是否具有统计学意义;(3)两组的Y值变化趋势有无统计学差异。

重复测量方差分析的软件实现如下:

```
data ex1;
input id group w0 w1 w2 w3 w4;
/*数据输入格式以每个时间点观测结局作为1个变量,
形成多变量结局*/
cards;
.....
;
proc glm data=ex1;
class group;
model w0-w4=group/nouni;
repeated time 5(0 1 2 3 4) contrast(1)/print summary;
/*time 指定时间点个数,如果时间点间隔不等,要在括号
中指定具体的时间点值,如1、2、4、7;如果时间点间隔相等(如本例),也可不指定具体时间点。contrast(1)表示以第
1次时间点为参照,其余时间点与参照时间点进行对比*/
manova h=group;
/*指定输出多变量的组间比较结果,如果不指定该语
句,默认不输出组间比较结果*/
run;
```

为了结果解释方便,首先在表1中列出了对照组和试验组在5个时间点的均值和标准差,图1则直观展示了两组的变化趋势。

绝大多数统计软件都会给出4个统计量:Wilks' lambda、Pillai's trace、Hotelling-Lawley trace 和 Roy's largest root。4个统计量的值虽然各不相同,但它们所对应的F值在绝大多数情况下都是一致的,因此实际中通常任选其一即可。以Wilks' lambda统计量为例,例1的主要结果见表2。

表1 接受试验药和对照药组患者在不同时间点的Y值($\bar{x} \pm s$)

组别	w0	w1	w2	w3	w4
对照组	105.4±4.0	106.8±4.2	105.7±5.1	107.5±4.6	108.6±6.0
试验组	109.6±11.0	111.0±10.6	113.4±8.7	118.0±9.8	121.0±6.3
合计	107.5±8.3	108.9±8.1	109.6±8.0	112.8±9.2	114.8±8.7

注:w0、w1、w2、w3、w4分别表示治疗前、治疗后1、2、3、4周




图1 两组在5个时间点的Y值

表2 重复测量方差分析结果

因素	Wilks' Lambda 值	F 值	P 值
group	0.344 5	5.33	0.006
time	0.350 6	6.95	0.002
group*time	0.567 7	2.86	0.061

表2结果显示,以0.05为检验水准,两组间差异有统计学意义($P=0.006$),5个时间点的差异有统计学意义($P=0.002$),组别和时间的交互效应无统计学意义。交互项反映了两条曲线是否平行,尽管从图1来看,两条曲线似乎并不平行,趋势变化似乎有差异,然而这种差异并无统计学意义。

上述SAS程序还同时给出了各个时间点的总体比较和分组比较结果:总体比较是指,以某一时间点为参照(程序中通过contrast指定以第1个时间点为参照),所有其他时间点与参照时间点的比较结果;分组比较是指,以某一时间点的两组差值为参照,所有其他时间点的两组差值与参照时间点两组差值的比较结果。表3给出了以第1个时间点为参照,其他时间点与参照点的总体比较和分组比较结果。

表3结果提示,从所有观测值来看(不分组),治疗后第3、4周与治疗前相比差异均有统计学意义(P 值均 <0.05)。从分组结果来看,治疗后第4周两组差值与治疗前两组差值的差异具有统计学意义($P=0.019$),结合表1可以看出,治疗前两组差值为4.2,治疗后第4周两组差值为12.4,两者差异有统计学意义。

(四)重复测量方差分析的局限性

(1)重复测量方差分析要求数据必须是均衡(balance)的,即每一时间点测量值的数量必须相等。这一点在实际

表3 重复测量方差分析各时间点的总体比较和
分组比较结果

比较时间点	均方	F 值	P 值
总体比较			
w1 vs w0	39.20	0.59	0.454
w2 vs w0	84.05	3.36	0.083
w3 vs w0	551.25	8.40	0.010
w4 vs w0	1 065.80	21.04	<0.001
分组比较			
w1 vs w0	0.00	0.00	1.000
w2 vs w0	61.25	2.45	0.135
w3 vs w0	198.45	3.02	0.099
w4 vs w0	336.2	6.64	0.019

注:w0、w1、w2、w3、w4分别表示治疗前、治疗后1、2、3、4周

中很难满足,例如,每组20人,分别在治疗前、治疗后1、3、6个月进行重复测量,很难保证每个人都能完成结果的观测;而且,即使都能完成,也难以保证这些人都恰好在1、3、6个月的固定时间点完成观测。对于某一被观测者,哪怕只有一个时间点的数据缺失或不符合时间方案,软件分析中该观测者的整条记录都会被删除。

(2)重复测量方差分析要求方差-协方差矩阵满足多元正态性和多元方差齐性。这两个条件,尤其是多元方差齐性,在实际中很容易被忽略,然而临床研究中却经常存在这种问题。例如,对高血压和非高血压两组人群在多个时间点进行重复测量,通常情况下,正常人群的收缩压往往变化不大,而血压高的人群,其收缩压往往变动更大,这就容易导致两组人群的方差-协方差矩阵方差不等。

(3)重复测量方差分析对时间变化趋势的分析不够充分。重复测量数据中,时间并不是以一个自变量的形式纳入模型分析,而是本身就作为模型的一部分。因此组别与时间的交互效应,难以深入分析,其变化趋势尽管可以通过正交多项式进行简单探索,但形式有限,不够灵活。例如,假定有试验组和对照组各4个时间点观测值,如果想要了解两组的每一时间点相对初始时间点的变化情况,重复测量方差分析就难以实现。

二、广义估计方程

(一)广义估计方程的思想

广义估计方程的计算过程很复杂,但思想却并不难理解。该方法假定在多次测量之间存在一定的相关结构(即作业相关矩阵),其思想就是把这种相关进行校正,然后得到校正后的参数估计值。

(二)广义估计方程中的作业相关矩阵

由于不同时间点观测之间的相关大小存在各种可能性,因此作业相关矩阵也有多种,常见的包括:

(1)独立结构(independence structure),即不同时间点上的测量值之间彼此独立,无相关关系。这种结构因为数据完全独立,实际上也无需考虑广义估计方程,直接采用常规的广义线性模型即可。

(2)等相关结构(exchangeable correlation structure),即假定任意两次观测之间的相关性是相等的,不随两个时间点之间的间隔大小而改变。

(3)一阶相关结构(one-dependent structure),表示某时间点的测量值只与其临近时间点的观测存在相关性,而与其他时间点的观测无关。例如,第2次观测只与第1次和第3次有相关,而与第4次无关。

(4)自相关(autocorrelation),即相关大小与间隔次数有关,相邻两次观测之间相关较强,间隔越远,相关性越小。例如,第2次观测与第1次和第3次观测相关性较大,与第4次观测的相关性较小。

(5)无结构相关(unstructured correlation),即假定不同时间点观测值的相关系数各不相同,不存在前面几种相关结构的规律。

作业相关矩阵的选择是广义估计方程中很关键的一部分,需要一定的统计学知识来判断。不少研究认为,作业相关矩阵的选择对参数估计结果的影响不大。然而实际数据分析中,指定不同的作业相关矩阵有时确实会产生不同的参数估计值和标准误。尽管一般差别不大,但笔者仍建议,尽量指定最为合适的作业相关矩阵,以获得最可靠的估计结果。

如何选择合适的作业相关矩阵,建议结合以下两种方式综合考虑:

(1)根据不同时间点观测值的相关系数矩阵考虑。如果任意两次的相关系数差不多,可考虑等相关;如果相关系数出现随时间间隔而规律性减小的趋势,可考虑自相关;如果无明显的规律,可考虑无结构相关。理论上,指定无结构相关最为稳妥,可以满足任意情形的相关系数矩阵,但它需要估计的参数也最多。例如,对于5次重复测量,如果指定等相关,只需要估计1个参数即可(只有1个相关系数);而无结构相关则需要估计任意两个时间点的相关系数,即10个参数,估计参数过多容易导致统计学效能(power)的降低。因此,实际分析中需要综合考虑,根据相关系数矩阵的提示选择较为合理的作业相关矩阵。

(2)结合QIC(quasi-likelihood under the independence model criterion)指标选择^[2]。QIC类似于广义线性模型的拟合优度指标AIC,只是最大似然值换成了准似然值,其值越小表示选择的作业相关矩阵越合适。与AIC指标类似,QIC指标中也有对变量的惩罚项,即QIC值不一定随着模型中变量的增多而变小,只有模型中含有意义的变量,其值才会变小,提示模型更优;如果纳入无意义的变量,其值反而会升高,提示模型变差。实际分析时,可以分别指定不同的作业相关矩阵,然后比较各自的QIC值,选择其中较小者。

(三)广义估计方程的用途

广义估计方程主要用于重复测量数据的分析,这里的重复测量不仅包括临床试验中较为固定、时间点较少的情形,也包括像生长发育监测、流行病学人群纵向观察等时间点较为灵活或时间点较多的情形。在临床试验的重复测量

数据分析中,广义估计方程也可以用于组间比较、时间点的比较、组间趋势变化的分析。在其他纵向观测数据中,广义估计方程可根据研究目的进行灵活分析。

(四)广义估计方程的SAS软件实现

广义估计方程的操作需要先进行一定的探索,确定作业相关矩阵。本例中我们分别指定了各种不同的作业相关矩阵,结果均一致,因此本例可任意指定一种作业相关矩阵,结果不受影响。简单起见,我们指定作业相关矩阵为等相关。

对例1数据采用基于等相关作业相关矩阵的广义估计方程,首先不加入时间与组别的交互项,先分析时间与组别各自的主效应。SAS程序如下:

```
data ex2;
input id group time y;
cards;
.....
;
proc gee data=ex2;
class id time/param=reference ref=first;
model y=time group;
repeated subject=id/within=time type=exch corrw;
/*subject 指定个体变量,重复测量数据中通常为个体的id编号;within 指定重复测量的变量,通常是时间点变量;type 指定作业相关矩阵;corrw 指定输出作业相关矩阵*/
run;
```

表4显示了组别与时间的主效应,结果提示,两组之间Y值评分差异有统计学意义($P=0.002$),治疗后第3周与治疗前差异有统计学意义($P=0.005$),治疗后第4周与治疗前差异有统计学意义($P<0.001$)。主效应是基于所有人(即不分组)的结果,因此,表4结果对应于重复测量方差分析表3结果中的总体比较(尽管结果并不完全一致)。参数估计值显示了差异情况,例如,group的参数估计值显示组间差异为7.8,即试验组的均值(114.6)与对照组的均值(106.8)相比高7.8;time 1 vs 0的参数估计值显示组间差异为1.4,提示第1周均值(108.9)比治疗前均值(107.5)高1.4。

如果分析中不加入时间与组别的交互项,相当于假定两条线是平行的,然而实际中这一假定并不一定满足。

表4 广义估计方程中组别与时间点的主效应估计结果

参数	估计值	标准误	Z值	P值
Intercept	103.600 0	1.507 8	68.71	<0.001
time 1 vs 0	1.400 0	1.735 5	0.81	0.420
time 2 vs 0	2.050 0	1.130 2	1.81	0.070
time 3 vs 0	5.250 0	1.857 2	2.83	0.005
time 4 vs 0	7.300 0	1.766 5	4.13	<0.001
group	7.800 0	2.479 8	3.15	0.002

注:time 1 vs 0、2 vs 0、3 vs 0、4 vs 0分别表示治疗后第1周、第2周、第3周、第4周与治疗前相比

图1可以看出两条线可能不平行,因此考虑在分析中纳入时间与组别的交互项,以便观察两组的变化趋势是否有差异。加入交互项的SAS程序如下:

```
data ex2;
input id group time y;
cards;
.....
;
proc gee data=ex2;
class id time/param=reference ref=first;
model y=time group time*group;
repeated subject=id/within=time type=exch corrw;
run;
```

表5显示了加入组别与时间交互效应的结果。一旦加入交互效应,组别与时间点反映的不再是主效应,而是单独效应。因此,如果了解组别与时间点的主效应,可以先不加入交互项。单独效应反映的不是所有人的估计结果,而是某一亚组(如对照组、第1周等)的估计结果。

单独效应的结果与变量赋值有很大关系,本例中试验组赋值为1,对照组赋值为0,时间点分别赋值为0~4。因此,表5中group反映的不是所有人两组的差值,而是治疗前这一时间点的两组差值(4.2);同样,time 1 vs 0反映的也不是所有人在第1周与治疗前的差值,而是对照组第1周与治疗前的差值(1.4)。

交互项的结果对应于重复测量方差分析表3结果中的分组比较。例如,group*time(1 vs 0)的参数估计值为0,它反映了第1周两组差值(4.2)与治疗前两组差值(4.2)的差值,也可以说,反映了试验组第1周-治疗前的值(1.4)与对照组第1周-治疗前的值(1.4)的差值,两种说法均可,取决于研究目的侧重说明什么。其他交互项的解释含义同此。

(五)广义估计方程分析的注意事项

(1)尽管广义估计方程需要考虑作业相关矩阵的设置,

表5 广义估计方程中组别、时间点及其交互效应的估计结果

参数	估计值	标准误	Z值	P值
Intercept	105.400 0	1.210 0	87.11	<.001
time 1 vs 0	1.400 0	0.862 6	1.62	0.105
time 2 vs 0	0.300 0	1.000 5	0.30	0.764
time 3 vs 0	2.100 0	1.053 1	1.99	0.046
time 4 vs 0	3.200 0	1.701 8	1.88	0.060
group	4.200 0	3.502 6	1.20	0.231
group*time(1 vs 0)	0.000 0	3.471 0	0.00	1.000
group*time(2 vs 0)	3.500 0	2.120 6	1.65	0.099
group*time(3 vs 0)	6.300 0	3.437 0	1.83	0.067
group*time(4 vs 0)	8.200 0	3.019 9	2.72	0.007

注:time 1 vs 0、2 vs 0、3 vs 0、4 vs 0分别表示治疗后第1周、第2周、第3周、第4周与治疗前相比

但绝大多数情况下,结果是一致的。建议实际分析中,首先可指定不同的作业相关矩阵,观察分析结果是否一致,如果一致,可以任选其一,否则可根据相关矩阵和 QIC 综合考虑,选择最合适的作业相关矩阵。

(2) 广义估计方程的结果比重复测量方差分析更接近模型的形式,因此不少非统计学专业人员可能对结果的解读存在一定困难,尤其是加入交互项的结果解读,需要仔细体会,否则很容易出现结果的解释错误。

(3) 广义估计方程比重复测量方差分析在分析思路上更为灵活,但这同时需要对统计学知识和软件操作的更高要求,因为广义估计方程的结果与自变量赋值有很大关系。例如对时间点赋值 0~4,与赋值为 1~5,二者给的结果会有不同。

三、多水平模型

多水平模型也被称为随机系数模型、随机效应模型、混合效应模型等,是一种处理非独立数据的方法。它是将个体与时间点分别看做水平 2 和水平 1 两个层次上的数据,水平 2 包含水平 1 (即每个人包含多次测量)。

(一) 基于重复测量数据的多水平模型思想

对于重复测量数据,一般线性模型的做法是将数据合并,估计自变量(如组别因素)的效应,这种做法忽略了个体间的变异,因为它假定个体之间是没有差异的。而现实情况是,数据的差异不仅体现在个体内(不同时间点),往往个体之间也是有差异的,因此分析时需要考虑个体间的差异。多水平模型正是基于这种思想,将个体间(水平 2)的变异分解出来,并估计其大小(通常用方差表示)。然后在此基础上,再估计自变量(如组别)的效应。

(二) 多水平模型中的随机效应

多水平模型不同于其他方法的特点就是可以估计随机效应。在重复测量数据中,随机效应可以理解为个体(水平 2)间的随机差异。常见的随机效应有两类,一是个体间只有截距存在差异,斜率相同(图 2 左),即每一个体高低不同,但时间变化趋势相同,此时可采用随机截距模型,估计截距之间的变异;二是个体间不仅截距存在差异,斜率也不相同(图 2 右),即每一个体不仅高低不同,而且每个人随时间变化的趋势也各不相同,此时可采用随机系数模型,同时估计截距和斜率之间的变异。重复测量数据分析中,通常先考虑随机截距模型,如果出于专业需要,或

经图形发现每个人的时间变化趋势差别较大,也可以考虑随机系数模型。

(三) 多水平模型中的用途

多水平模型用途十分广泛,它可以用于具有层次结构的数据分析,如“学校-学生”“乡镇-村民”“患者-肿瘤部位”“个体-时间”等数据,重复测量数据可以看做是多水平数据的特例。

(四) 多水平模型 SAS 软件实现

多水平模型的分析通常需要进行一定的初步探索过程,然后才可考虑最终确定的模型。重复测量数据的多水平模型的分析,建议第一步先绘制每一个体随时间的变化趋势图,根据图形为进一步的分析提供参考。例 1 中 20 名个体随时间变化的趋势图见图 3,可以看出,每个人的 Y 值高低不同,差异较大,因此至少可以考虑随机截距模型。从时间变化的斜率来看,尽管在某些个体中起伏较大,但总的大都是呈增加趋势,而且可以发现,试验组增加的趋势似乎幅度更大。如果每个人随时间变化趋势基本一致,拟合随机截距模型即可,否则需要考虑随机系数模型。当然图形只是给我们一个初步的主观判断,是否需要采用随机系数模型,可结合统计分析结果来确定。




图 3 20 名患者的症状评分指标 Y 随时间的变化趋势图

根据图 3 提示,我们先考虑拟合随机截距模型,以确定个体间是否变异较大。实际中通常先拟合一个随机截距的空模型(即模型中不包含任何变量),以观察个体间变异情况。SAS 程序如下:

```
data ex2;
input id group time y;
cards;
.....;
proc mixed covtest data=ex2;
class id;
model y=/solution;
random int/subject=id type=un solution;
/*random 指定随机效应,此处指定 int
(截距),拟合随机截距模型;type 指定水
```




图 2 随机效应示例图

平 2 变量,重复测量数据中即个体 id;type 指定协方差结构,常见为 vc 和 un;solution 指定输出随机效应估计结果*/

```
run;
```

表 6 显示了随机效应估计结果,其中 id 对应的估计值反映了个体间的变异大小(即水平 2 的方差),Z 值和 P 值反映了方差与 0 相比的统计学检验结果,本例分析结果显示,个体间差异较大,且这种变异有统计学意义;residual 则显示了水平 1 的误差,即个体间变异外的随机误差。

表 6 随机截距模型(空模型)的随机效应估计结果

协方差参数	对象	估计值	标准误	Z 值	P 值
UN(1, 1)	id	41.023 8	15.738 5	2.61	0.005
Residual		36.765 0	5.813 1	6.32	<0.001

这里需要提醒一点,由于此处估计的是方差,方差不可能小于 0,因此此处的 P 值是单侧 P 值。有的统计软件给出的是双侧 P 值,是有问题的。

通过对空模型的分析,可以看出,个体间的确存在较大的变异,且有统计学意义,因此需要考虑多水平模型。如果个体间变异很小,无统计学意义,也可以不用多水平模型,直接采用普通的一般线性模型即可。

基于表 6 结果,进一步拟合随机截距模型。此时的分析与广义估计方程类似,如果要考虑组别和时间的主效应,可先不加入交互项。SAS 程序如下:

```
proc mixed covtest data=ex2;
class id time(ref=first);
model y=time group/solution;
random int/subject=id type=un solution;
run;
```

结果显示,时间和组别的主效应均有统计学意义(表 7)。与表 4 的广义估计方程结构相比,两种方法的参数估计值相同,但标准误不同,从而导致不同的统计量和 P 值。但总的来说,结论是一致的。二者的解释也相同,这里不再赘述。

表 7 随机截距模型中组别与时间点的主效应估计

参数	参数估计值	标准误	Z 值	P 值
Intercept	103.600 0	2.141 8	48.37	<0.001
time 1 vs 0	1.400 0	1.711 0	0.82	0.416
time 2 vs 0	2.050 0	1.711 0	1.20	0.235
time 3 vs 0	5.250 0	1.711 0	3.07	0.003
time 4 vs 0	7.300 0	1.711 0	4.27	<0.001
group	7.800 0	2.614 0	2.98	0.004

此时拟合的是基于随机截距模型的结果,我们进一步考虑是否有必要采用随机系数模型,即每个人随时间变化的趋势(每个人的斜率)是否有差异。SAS 程序如下:

```
data ex2;
input id group time y;
timec=time;
```

```
cards;
```

```
.....
```

```
;
```

```
proc mixed covtest data=ex2;
```

```
class id time(ref=first);
```

```
model y=group time/solution;
```

```
random int timec/subject=id type=un solution;
```

/*random 指定 int 和 timec 作为随机效应,拟合随机系数模型 */

```
run;
```

注意程序中新产生了 1 个新变量 timec,其值完全等同于 time,即 0~4。产生的目的主要是因为 random 语句指定的随机效应通常为连续变量而不是分类变量(否则容易导致估计不出结果),然而固定效应估计中我们期望将其作为分类变量,这样才能发现每一时间点与参照时间点的差异。因此分析时将 time 作为分类变量,放在固定效应的估计中,而 timec 作为连续变量放在随机效应的估计中。当然,如果固定效应分析本身就将 time 作为连续变量,就无需这一步。

表 8 给出了随机系数模型的随机效应估计结果,与表 6 相比,尽管只增加了 1 个随机斜率,但结果却增加了 2 个,除了时间的随机效应外,还有 1 个是时间随机效应与截距随机效应的协方差。

表 8 随机系数模型的随机效应估计结果

协方差参数	对象	估计值	标准误	Z 值	P 值
UN(1, 1)	id	54.427 0	26.468 8	2.06	0.020
UN(2, 1)	id	-6.633 8	5.033 2	-1.32	0.188
UN(2, 2)	id	0.820 1	1.257 4	0.65	0.257
Residual		27.224 7	5.099 6	5.34	<0.001

表 8 结果中,第一行反映的是截距的方差,提示个体间差异较大,且有统计学意义($P=0.020$);第三行反映的是(时间变量的)斜率的方差,提示每一个体随时间变化的斜率差异并不大,且无统计学意义($P=0.257$);第二行是截距和斜率的协方差,这一结果很有实际意义,它反映了截距与斜率的相关性,表 8 结果中,该估计值为负数,提示如果个体初始时间点的值较大,其斜率较小,反之,如果某个人初始时间点的值较低,其斜率变化则较大。通俗地说,如果某人一开始 Y 值很低,那么后期很可能会增长(或降低)很快,而如果一开始 Y 值较高,后期增长(或降低)速度会较慢。当然由于该值并无统计学意义($P=0.188$),也可能并无这种关联。

表 8 结果提示,time 作为随机效应并无统计学意义,因此实际中考虑随机截距模型即可。

进一步在模型中加入交互项,以考量两组随时间变化的趋势是否相等。SAS 程序如下:

```
proc mixed covtest data=ex2;
class id time(ref=first);
model y=time group time*group/solution;
random int/subject=id type=un solution;
```


run;

估计结果及其解释仍与广义估计方程类似,结论也是一致的(表9),此处不再重复解释。

表9 随机截距模型中组别、时间点及其交互效应估计

参数	参数估计值	标准误	Z值	P值
Intercept	105.40	2.363 6	44.59	<0.001
time 1 vs 0	1.400 0	2.329 2	0.60	0.550
time 2 vs 0	0.300 0	2.329 2	0.13	0.898
time 3 vs 0	2.100 0	2.329 2	0.90	0.370
time 4 vs 0	3.200 0	2.329 2	1.37	0.174
group	4.200 0	3.342 6	1.26	0.213
group*time(1 vs 0)	-1.81×10 ⁻¹³	3.293 9	0.00	1.000
group*time(2 vs 0)	3.500 0	3.293 9	1.06	0.292
group*time(3 vs 0)	6.300 0	3.293 9	1.91	0.060
group*time(4 vs 0)	8.200 0	3.293 9	2.49	0.015

(五)多水平模型分析的注意事项

多水平模型分析的关键是如何指定随机效应,不少非统计专业人员难以理解这一含义。从实际分析的角度来看,建议一定要先绘制每一个体随时间变化的折线图,根据图形大致观察每一个体是否高低不同或变化趋势不同。如果只是高低不同,但每一折线的趋势基本类似(如都呈上升趋势),此时可考虑随机截距模型;如果不仅高低不同,而且每一折线的趋势也不同(如有的呈上升趋势,有的呈下降趋势),此时可考虑随机系数模型。从软件分析的角度来讲,随机截距模型即在软件中指定截距为随机效应,随机系数模型即在软件中指定截距和时间变量作为随机效应,作为随机效应的时间变量建议为连续变量。

另外,需要注意多水平模型的结果解读:多水平模型的固定效应分析与一般线性模型相同,在重复测量数据中,可根据实际需要指定组别变量、时间变量、组别与时间的交互项,其结果解释与传统的一般线性模型也都相同。随机效应的估计结果与固定效应不同,它反映的是变异大小,如果只有截距是随机效应,给出的是个体间截距的变异大小;如果截距和时间变量均作为随机效应,给出的是个体间截距的变异大小和个体间(随时间变化)斜率的变异大小。

四、讨论

(一)三种方法的比较与选择

(1)从方法的复杂程度来看,重复测量方差分析最为简单,分析程序或操作过程较为固定,且结果与普通的方差分析类似,是非统计专业人员常用的方法;广义估计方程难易程度中等,软件实现过程大多数都是固定的语法,如果不考虑作业相关矩阵的选择,其软件实现和结果解读仍较为简单,对于非统计人员来说,可以作为重复测量方差分析的补充;多水平模型最为复杂,其软件实现需要较多的专业知识,通常由专业的统计学人员完成。

(2)从应用场景来看,如果是基于时间点的重复测量,重复测量方差分析、广义估计方程和多水平模型均可考虑;

如果是基于空间的重复测量,如同一个家庭的不同成员、同一窝的仔鼠、同一患者的不同肿瘤部位等,建议首选多水平模型。

(3)从结局的类型来看,如果结局是连续变量,重复测量方差分析、广义估计方程和多水平模型均可考虑;如果结局是其他类型变量,如分类变量、计数变量等,可以考虑广义估计方程或多水平模型。

(4)从分析目的和自变量类型来看,如果只是比较组间差异,即自变量只有组别因素(二分类或多分类),重复测量方差分析、广义估计方程和多水平模型均可考虑;如果是分析多个自变量对结局的影响,可考虑广义估计方程或多水平模型。

(5)从数据的均衡性来看,如果时间点固定,且每个时间点的数据都无缺失,采用重复测量方差分析、广义估计方程或多水平模型均可;如果时间点不固定,如有的人观测时间点是1、3、6,有的是1、4、7,可考虑广义估计方程或多水平模型。

(6)从观测次数来看,如果观测次数较少,如临床试验的重复观测通常3~5个时间点,此时重复测量方差分析、广义估计方程和多水平模型均可考虑;如果观测次数较多,如生长发育监测数据,往往可能有十多次观测值,建议考虑多水平模型或广义估计方程,可以灵活地探索各种可能的变化趋势。

(7)关于广义估计方程和多水平模型的选择。广义估计方程和多水平模型的参数估计值通常差别不大,事实上,在某些特定情况下,它们的结果是完全一致的^[3]。但它们的分析侧重不同,广义估计方程主要是考虑到各次测量之间的相关性,将其作为作业相关矩阵进行校正;多水平模型侧重个体(水平2)间的变异分解,将其作为随机效应进行估计。总的来说,多水平模型估计的信息更丰富,它可以提示个体间的变异是如何变化的。但如果研究目的不追求这么多,只是关注固定效应(如组别因素)的估计结果,两种方法均可考虑。

(8)综合建议,仅从方法本身来看,建议首选多水平模型,该法提供的信息最多,可以提供特定的变异分解结果,结合专业知识给出详细解释,其应用范围也最广;其次建议选择广义估计方程,该法尽管结果不如多水平模型提供的详细,但在时间等自变量的分析方面,跟多水平模型一样具有灵活性;最后才考虑选择重复测量方差分析,该法相对而言缺乏灵活性,应用范围较窄,实际中主要用于目的较为单纯的临床试验重复测量数据。

(二)关于重复测量数据中的缺失值问题

(1)在重复测量数据中,数据缺失导致的最明显问题就是不平衡。不平衡这一问题对重复测量方差分析影响最大,对广义估计方程和多水平模型并无影响。注意这里说的是数据不平衡对广义估计方程和多水平模型并无影响,并不等同于数据缺失对这两种方法无影响。

(2)数据缺失的影响需要分别看待。数据缺失类型一

般可分为三种^[4]:完全随机缺失(MCAR)、随机缺失(MAR)和非随机缺失(NMAR)。完全随机缺失是指缺失与观察结局或其他协变量均无关,如研究对象死亡、地址搬迁等,这些都是不可控的随机因素;随机缺失是指缺失与某协变量有关,如与性别有关(如男性缺失较多而女性缺失较少)、与干预有关(如试验组缺失较多而对照组缺失较少);非随机缺失是指与研究结局自身有关的缺失,如研究降压药的降压效果,如果研究对象因为觉得降压效果不好而退出,即降压效果不好的人缺失多,降压效果好的人缺失少,这就属于非随机缺失。

对于重复测量方差分析而言,任何一种数据缺失都会对结果造成影响,影响大小取决于具体的缺失情形;广义估计方程是基于完全随机缺失的假设^[5-6],因此完全随机缺失模式对广义估计方程的结果影响不大,此时其参数估计值仍是稳定的;对于随机缺失模型,如果要采用广义估计方程,可考虑加权的广义估计方程(Weighted Generalized Estimating Equations),该法是基于随机缺失的假定,但仅限于失访模式(即一个人在某个时间点缺失后,后面的时间点均无数据)^[7-8];多水平模型是基于随机缺失的假定^[9],因此随机缺失模式对多水平模型的估计结果影响不大。对于所有的三种方法,如果是非随机缺失,仍会对结果造成影响。

因此,重复测量方差分析建议应用于无缺失的情形;如果缺失是完全随机的,此时广义估计方程和多水平模型均可考虑;如果是随机缺失,建议选择多水平模型,如果是典型的失访且失访后再未进入研究,也可考虑加权的广义估计方程;如果是非随机缺失,此时结果必然会受影响,可补救的措施是缺失值填补,然而这并不意味着填补后的结果是可靠的^[10],只能说是作为一种可考虑的补救方式。

(三)关于结局变量的正态性

理论上,对于定量资料,重复测量方差分析、广义估计方程和多水平模型都要求因变量满足正态分布,但由于重复测量方差分析是将多个时间点的观察值作为多元指标,因此需要满足多元正态性。多元正态分布的检验在 SAS 中可通过宏%multnorm实现,可从 SAS 官方网站免费下载。广义估计方程和多水平模型是将结局作为单变量,因此进行常规的正态性检验即可。

如果结局不满足正态性,目前并无太理想的处理方式,最常见的仍是变量变换(如取对数变换、秩变换)。当然变量变换也有一定缺陷,最大的问题就是结果的解释。但相比之下,变量变换仍是利大于弊。因为这几种方法的参数估计方法(如最大似然估计)都是基于正态分布假定,如果不满足基本假定,估计结果及最终的 P 值必然有偏。相比之下,变量变换后如果满足正态分布,尽管变量测量尺度发生变化,但统计分析结果是正确的。

(四)关于时间变量纳入的尺度

在重复测量方差分析中,时间含在因变量之中,因此分析时无需再次指定时间变量。但在广义估计方程和多水平模型中,时间变量是单独的自变量,以何种形式纳入模型是

一个较为关键的问题。

本文的分析中,一直是将时间变量作为分类变量,并以虚拟变量的形式纳入模型,即指定治疗前为参照,其余时间点与治疗前进行比较。这种纳入形式是最为稳妥的方式,不管结局随时间变化趋势是何种形式,均可较好地展示出变化规律。但从实际角度来看,并不是所有分析都需要采用这种方式。

如果从图形中大致能够判断出曲线的变化趋势(如大致为直线上升),可以直接将时间作为连续变量纳入模型。与分类变量相比,作为连续变量纳入可以减少需要估计的参数个数,这在样本例数较少时尤为重要。如本文的分析案例中,从图 1 来看,大致呈直线上升趋势,因此也可以考虑直接作为连续变量纳入,估计结果差别并不大。

多水平模型中,如果要估计时间点对结局影响的随机效应(即假定每个人随时间变化趋势不同,需要估计个体间斜率的变异),此时建议将时间作为连续变量而不是分类变量,否则容易导致无法估计。如果固定效应仍需要将时间作为分类变量,此时可指定一个新变量,其值等同于原来的 time 变量,但将其作为分类变量纳入固定效应分析。

如果重复观测时间点较多,如重复观察 10 多次,此时也不建议将时间变量作为分类变量,而是将其作为连续变量纳入,并根据图形大致形状,指定相应的多项式来拟合变化趋势。如果作为虚拟变量,拟合结果太细,反而容易忽略了真正的变化规律。

利益冲突 所有作者均声明不存在利益冲突

参 考 文 献

- [1] Haase RF. 多元广义线性模型[M]. 臧晓露, 译. 上海: 格式出版社, 2017.
- [2] Pan W. Akaike's information criterion in generalized estimating equations[J]. Biometrics, 2001, 57:120-125.
- [3] Twisk JWR. Longitudinal data analysis. A comparison between generalized estimating equations and random coefficient analysis[J]. Eur J Epidemiol, 2004, 19: 769-776.
- [4] Allison PD. 缺失数据[M]. 林毓玲, 译. 上海: 格式出版社, 2012.
- [5] Omar RZ, Wright EM, Turner RM, et al. Analysing repeated measurements data: a practical comparison of methods Statist[J]. Stat Med, 1999; 18(13): 1587-1603. DOI: 10.1002/(sici)1097-0258(19990715)18: 13<1587::aid-sim141>3.0.co;2-z
- [6] Albert PS. Tutorial in biostatistics longitudinal data analysis (repeated measures) in clinical trials[J]. Stat Med, 1999, 18, 1707-1732.
- [7] Fitzmaurice GM, Laird NM, Ware JH. Applied Longitudinal Analysis[M]. 2nd ed. Hoboken, NJ: John Wiley & Sons, 2011.
- [8] Preisser JS, Lohman KK, Rathouz PJ. Performance of weighted estimating equations for longitudinal binary data with drop-outs missing at random[J]. Stat Med, 2002, 21(20):3035-3054. DOI: 10.1002/sim.1241.
- [9] 王济川, 谢海义, 姜宝法. 多层统计分析模型——方法与应用[M]. 北京: 高等教育出版社, 2008.

- [10] Twisk JWR. Applied longitudinal data analysis for epidemiology: a practical guide[M]. Cambridge: Cam-bridge University Press, 2003.

单选题 [完成以下单选题可获得 II 类继续教育学分(答题二维码及 II 类学分授予说明见活插页)]

1. 两组重复测量数据的比较分析中,以下不恰当的做法是:
 - A. 采用重复测量方差分析
 - B. 采用广义估计方程
 - C. 采用多水平模型
 - D. 采用 t 检验对每个时间点分别比较
2. 重复测量方差分析不能用于:
 - A. 两组重复测量数据的比较
 - B. 多次测量时间点之间的比较
 - C. 两组测量的变化趋势比较
 - D. 估计两组的 OR 值
3. 下面说法错误的是:
 - A. 重复测量方差分析可用于两组分类资料的比较
 - B. 广义估计方程可用于纵向资料的分析
 - C. 多水平模型可用于两组重复测量数据的趋势分析
 - D. 重复测量方差分析需要满足多元正态性
4. 下面哪种方法受缺失值的影响最大:
 - A. 重复测量方差分析
 - B. 广义估计方程
 - C. 多水平模型
 - D. 加权广义估计方程
5. 下面关于多水平模型说法错误的是:
 - A. 多水平模型可用于组间重复测量数据的分析
 - B. 多水平模型可用于多层次数据的分析
 - C. 多水平模型只能用于时间重复测量数据的分析
 - D. 多水平模型可用于生长发育多次测量数据的分析



中华医学会